

Model Pembelajaran

Grup Siklis dan Dihedral dengan Maple



oleh: *Sahid*

Laboratorium Komputer Jurdik Matematika FMIPA UNY

- **Definisi Grup Siklis dan Dihedral ***

Grup siklis (Herstein, 1975: 30):


Misalkan a sebaran simbol dan himpunan C_n terdiri atas semua simbol a^i , $i=0, 1, 2, \dots, (n-1)$, dengan aturan $a^0 = e$, $a^i \cdot a^j = a^{(i+j)}$ jika $i+j < n$ dan $a^i \cdot a^j = a^{(i+j-n)}$ jika $i+j > n$. C_n dengan operasi perkalian tersebut dinamakan grup siklis berorder n yang dihasilkan oleh a dengan elemen idenitas e .

Grup dihedral (Herstein, 1975: 54):

Misalkan r dan R adalah dua simbol yang berbeda dan himpunan D_n terdiri atas semua simbol $r^i R^j$, untuk $i, j = 0, 1, 2, \dots, (n-1)$ dengan aturan (1) $r^i R^j = r^k R^l$ jika dan hanya jika $i=k$ dan $j=l$, (2) $r^2 = R^n = e$, untuk $n > 2$, dan (3) $r \cdot R = R^{(n-1)} r$. D_n dengan operasi perkalian tersebut dinamakan grup dihedral berorder $2n$ yang dihasilkan oleh r dan R dengan elemen idenitas e .

• Realisasi Grup Siklis dan Dihedral *

- Grup siklis dapat disajikan sebagai himpunan **rotasi** segi banyak (n) beraturan, sejauh kelipatan ($360^\circ / n$).
- Grup dihedral dapat disajikan sebagai himpunan simetri (**rotasi dan refleksi/ pencerminan**) segi banyak beraturan.
- Setiap titik sudut segi- n ditandai dengan angka $1, 2, \dots, n$, secara berlawanan arah dengan arah jarum jam.
- Baik rotasi maupun refleksi dapat dipandang sebagai transformasi yang memindahkan label titik-titik sudut tersebut ke titik-titik sudut yang lain.



Gambar 1 Contoh Rotasi dan Refleksi pada Segi-5 Beraturan

Prosedur Maple untuk Menghasilkan Grup Siklis dan Dihedral

- ❖ Setiap rotasi/refleksi segi- n dinyatakan sbg. permutasi $[a_1, a_2, a_3, \dots, a_n]$, yang berarti pemetaan 1 ke a_1 , 2 ke a_2 , 3 ke a_3 , ..., n ke a_n dengan a_k di dalam $\{1, 2, 3, \dots, n\}$, $k = 1, 2, 3, \dots, n$.
- ❖ **Rotasi** dalam segi- n : $[i, i+1, \dots, n, 1, 2, \dots, i-1]$, $i=1, 2, \dots, n$.
- ❖ **Refleksi (simetri) segi- n** : $[n+1-i, n-i, \dots, 1, n, n-1, \dots, n-i+2]$, $i=1, 2, \dots, n$
- ❖ Dalam Maple permutasi $[k, k+1, k+2, \dots, n, 1, 2, \dots, k-1]$ dapat dituliskan sebagai $[\$k..n, \$1..k-1]$.
- ❖ Fungsi Maple sebagai berikut untuk menghasilkan elemen-elemen (dalam bentuk permutasi) grup siklis dan dihedral pada segi- n (lihat Mihailovs, 2001).
 - > **siklis:=n->[seq([\$i..n, \$1..i-1], i=1..n)]:**
 - > **dihedral:=n->[op(siklis(n)), seq([n+1-j\$j=i..n, n+1-j\$j=1..i-1], i=1..n)]:**

• Contoh Menghasilkan Elemen-elemen Grup Siklis/Dihedral

Grup rotasi segitiga beraturan:

> **siklis(3);**

[[1, 2, 3], [2, 3, 1], [3, 1, 2]]

Grup simetri segitiga beraturan:

> **dihedral(3);**

[[1, 2, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1], [2, 1, 3], [1, 3, 2]]

Grup simetri bujursangkar:

> **dihedral(4);**

[[1, 2, 3, 4], [2, 3, 4, 1], [3, 4, 1, 2], [4, 1, 2, 3], [4, 3, 2, 1], [3, 2, 1, 4], [2, 1, 4, 3],
[1, 4, 3, 2]]

- ❖ Setiap elemen dapat dinyatakan dengan bilangan-bilangan ordinalnya (urutannya) dalam grup tersebut.
- ❖ Invers suatu elemen (dengan notasi baru ini), dicari dgn prosedur berikut (Mihailovs, 2001):

```
> invep:=proc(a,G) local i,v,b,k;  
    k:=nops(G[a]); b:=[0$k];  
    for i to k do b[G[a][i]]:=i od;  
    member(b,G,'v'); v  
end:
```

Contoh:

Invers elemen ke-5 dalam grup siklis segi-12 adalah:

> **invep(5,siklis(12));**

9

yakni elemen ke-9.

Invers elemen ke-5 dalam D_4 adalah:

> **invep(5,dihedral(4));**

5

yakni elemen ke-5.

• Visualisasi Grup Siklis dan Grup Dihedral*

Prosedur **Gambar** berikut ini berguna untuk menampilkan grafik ks elemen dari grup G , mulai dari elemen ke- m , ditampilkan dalam k kolom dan s baris.

> **with(plots) : with(plottools) :**

```
> Gambar:=proc(G,m,k,s) local segin,seginasli,seginhasil,
n,a,b,i,j1,j2,p,panah,l,label,titikasal,titikhasil,nomor;
n:=nops(G[1]);# (a,b) adalah pusat segi-n beraturan
segin := (a,b)-> [seq([a+cos(2*Pi*i/n),b+sin(2*Pi*i/n)],
i=1..n)]:
```

```
seginasli:=seq(seq(segin(10*j1,-4*j2),j1=1..k),j2=1..s);
```

```
seginhasil:=seq(seq(segin(10*j1+4.5,-4*j2),j1=1..k),
j2=1..s);
```

```
p:=polygonplot({seginasli,seginhasil},axes=NONE,
scaling=CONSTRAINED,color=aquamarine):
```

```
panah:=seq(seq(arrow([10*j1+1.6,-4*j2],
vector([1.2,0]),.05,.3,.3,color=blue),j1=1..k),j2=1..s):
```

```
label:=(a,b,l)-> seq([a+1.4*cos(2*Pi*i/n),
b+1.4*sin(2*Pi*i/n),l[i+1]],i=0..n-1):
```

```
titikasal:=textplot([seq(seq(label(10*j1,-4*j2,$1..n)),
j1=1..k),j2=1..s)],color=black):
```


```
titikhasil:=textplot([seq(seq(label(10*j1+4.5,-4*j2,
G[invep(m-1+j1+k*(j2-1),G)]),j1=1..k),j2=1..s)],color=red):
```

```
nomor:=textplot([seq(seq([10*j1+2.2,-4*j2+.5,
m-1+j1+k*(j2-1)],j1=1..k),j2=1..s)],color=blue):
```

```
display(p,panah,titikasal,titikhasil,nomor)
```


```
end:
```

> **Gambar(siklis(4),1,2,2);**




Gambar 2. Visualisasi grup rotasi bujur sangkar (C_4)

> **Gambar (dihedral (4) , 1, 2, 4) ;**



Gambar 3. Visualisasi grup simetri bujur sangkar (D_4)

> **Gambar (dihedral (10) , 11, 1, 1) ;**



Gambar 5. Visualisasi sebuah elemen grup simetri segi-10 (D_{10})

> **dihedral (10) [11] ;**
[10, 9, 8, 7, 6, 5, 4, 3, 2, 1]

• **Tabel Cayley dan Pengujian Grup Abelian ***

❖ Fungsi **kalip** untuk mengalikan dua buah elemen a dan b (a dan b menyatakan urutan anggota) dari himpunan atau grup G .

> **kalip:=proc(a,b,G) local i,v;**
member([seq(G[a][G[b][i]],i=1..nops(G[a])),G,'v');v end;

Contoh:

> **kalip(7,3,dihedral(4));**

5

> **dihedral(4)[7]; # Elemen ke-7**

[2, 1, 4, 3]

> **dihedral(4)[3]; # Elemen ke-3**

[3, 4, 1, 2]

> **dihedral(4)[5]; # Elemen ke-5**

[4, 3, 2, 1]

> **dihedral(4)[kalip(7,3,dihedral(4))];**

[4, 3, 2, 1]

❖ Prosedur Maple **cayley** untuk menyusun tabel *Cayley* sebuah grup

> **cayley:=G->Matrix(nops(G), (i,j)->kalip(i,j,G)):**

Contoh :

Tabel *Cayley* untuk D_3 adalah:

> **d[3]:=cayley(dihedral(3));**

$$d_3 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ 2 & 3 & 1 & 6 & 4 & 5 \\ 3 & 1 & 2 & 5 & 6 & 4 \\ 4 & 5 & 6 & 1 & 2 & 3 \\ 5 & 6 & 4 & 3 & 1 & 2 \\ 6 & 4 & 5 & 2 & 3 & 1 \end{bmatrix}$$

• Menguji grup Abelian:

> **isAbelian:=G->IsMatrixShape(cayley(G), symmetric):**

Contoh:

> **isAbelian(dihedral(3));**

false

Tabel *Cayley* grup siklis bersifat sangat simetris:

> **C[5]:=cayley(siklis(5));**

$$C_5 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 2 & 3 & 4 & 5 & 1 \\ 3 & 4 & 5 & 1 & 2 \\ 4 & 5 & 1 & 2 & 3 \\ 5 & 1 & 2 & 3 & 4 \end{bmatrix}$$

Jadi grup C_5 adalah grup Abelian:

```
> isAbelian(siklis(5));
```

true

Operasi Perkalian dan Inversnya

Operator perkalian pada suatu **Grup** didefinisikan dengan menggunakan fungsi **kalip**:

```
> `&*` := (a,b) -> kalip(a,b,Grup);
```

Invers elemen suatu **Grup** didefinisikan dengan menggunakan fungsi **invep**:

```
> `&-` := a -> invep(a, Grup);
```

Contoh:

```
> Grup := dihedral(3);
```

```
> 3&*4;
```

5

Langsung menggunakan perintah **kalip**.

```
> kalip(3,4,Grup);
```

5

```
> &-2;
```

3

```
> &-2&*2;4&*2;
```

1

5

```
> 2&*5&* &-2;
```

6

Pembahasan

- ❖ *Doing mathematics with computer; Learning mathematics with computer; Teaching mathematics with computer. ???*
- ❖ *Ya, apabila sudah melek komputer (computer literate).*
- ❖ Dengan menggunakan Maple, komputasi dan visualisasi serta eksplorasi sifat-sifat grup siklis dan dihedral dapat dilakukan secara mudah dan cepat. Perintah dan program sangat sederhana.
- ❖ Setelah menuliskan fungsi/prosedur Maple yang diperlukan, mahasiswa dapat melakukan eksplorasi sendiri serta menyelesaikan soal-soal atau masalah tentang grup siklis dan grup dihedral dengan menggunakan Maple. Ditinjau dari teori belajar, hal ini sesuai dengan prinsip belajar dengan mengerjakan (*learning by doing*) dan akan memperkuat penguasaan konsep (Arsham, 2002).
- ❖ Kemampuan dan kemudahan-kemudahan yang tersedia pada Maple menjadikannya sebuah alat yang handal dan fleksibel bagi para penggunanya di bidang pendidikan, riset, dan industri. Tidak sekedar itu, bahkan Maple juga merupakan alat belajar dan mengajar yang menakjubkan (Moore, 2001: 9).
- ❖ Maple bukan merupakan software CAI (*Computer-Assisted Instruction*)^o, namun dapat digunakan sebagai CBE (*Computer-Based Education*) dan CBI (*Computer-Based Instruction*). CBE dan CBI memiliki pengertian yang lebih luas, mencakup semua penggunaan komputer untuk pendidikan, baik dalam bentuk aktivitas dengan komputer secara terpisah maupun aktivitas dengan komputer untuk memperkuat penguasaan materi yang diberikan oleh guru (Cotton, 2001).
- ❖ Kegiatan pembelajaran ini dapat diberikan dalam bentuk praktikum yang dapat dilakukan oleh mahasiswa secara mandiri atau dengan bimbingan dosen.
- ❖ **Maple** sebagai alat riset di kalangan perguruan tinggi dan industri, sebagai alat pembelajaran, khususnya untuk bidang matematika, fisika, dan teknik, di sekolah dan perguruan tinggi, (Waterloo Maple, 1998-99:4; Moore, 2001; Waterloo Maple, 2001b; Waterloo Maple, 2002:4).
- ❖ Melalui model pembelajaran ini, mahasiswa akan memperoleh setidaknya dua hal, yakni: (1) pemahaman konsep grup siklis dan grup dihedral yang lebih mendalam, dan (2) ketrampilan menggunakan komputer, khususnya software Maple, untuk alat bantu pemecahan masalah matematika

❖ Beberapa keuntungan model pembelajaran ini:

1. tidak memerlukan kemampuan pemrograman yang rumit;
2. perintah-perintah Maple yang digunakan cukup sederhana;
3. baik dosen maupun mahasiswa tidak perlu melakukan pekerjaan manual baik, untuk komputasi maupun visualisasi;
4. visualisasi grup siklis dan dihedral dapat dilakukan dengan mudah sesuai kehendak dan kebutuhan;
5. elemen-elemen suatu grup dapat dinyatakan secara visual, notasi permutasi maupun angka, sesuai kebutuhan;
6. komputasi dapat dilakukan untuk sebarang grup siklis dan dihedral, baik yang berorder kecil maupun besar;
7. pembuatan tabel Cayley dapat dilakukan secara mudah dan sangat cepat;
8. mahasiswa mempunyai kesempatan banyak untuk melakukan eksplorasi dan penyelidikan sifat-sifat suatu grup siklis dan dihedral;
9. operasi perkalian dalam suatu grup dapat dilakukan dengan menggunakan fungsi maupun dengan operator, seperti dalam matematika;
10. penyelidikan sifat-sifat yang lebih lanjut dapat dilakukan dengan menggunakan fungsi dan prosedur Maple yang sudah ada.

*) CAI merupakan istilah yang lebih sempit dan lebih merujuk kepada penggunaan komputer untuk kegiatan drill dan praktek, tutorial, simulasi, atau permainan, baik tanpa atau dengan bimbingan guru (Cotton, 2001; Doerr, 1979:121; Hopper & Mandell, 1987: 212). Cotton (2001) juga menyebutkan dua istilah lain yang terkait adalah CMI (*Computer-managed instruction*) dan CEI (*Computer-enriched instruction*). CAI merujuk pada pemakaian komputer untuk mengelola data dan mengambil keputusan instruksional, mengevaluasi hasil belajar siswa dan mengarahkan mereka kepada sumber belajar yang sesuai, serta menyimpan catatan kemajuan belajar mereka. CEI adalah aktivitas belajar dengan komputer di mana komputer melakukan (1) menghasilkan data sesuai permintaan siswa untuk melukiskan hubungan di dalam model realitas sosial atau fisik, (2) menjalankan program yang dibuat siswa, atau (3) memberikan pengayaan umum dalam bentuk latihan yang relatif tak terstruktur yang dirancang untuk merangsang dan memotivasi siswa.

Daftar Pustaka

Arsham, Hossein. 2002. *Computer-assisted Learning ,Concepts & Techniques*.
<http://ubmail.ubalt.edu/~harsham/opre640a/partX.htm>

Cotton , Kathleen. 2001. *Computer-Assisted Instruction*. Northwest Regional Educational Laboratory. <http://www.nwrel.org/scpd/sirs/5/cu10.html>

Doerr, Christine. 1979. *Microcomputer and the 3R's, A Guide for Teachers*. New Jersey, Hayden Book Company, Inc.

Herstein, I.N. 1975. *Topics in Algebra*. second edition. Singapore, John Wiley & Sons.

Mihailovs, Alec. 2001. *Abstract Algebra with Maple*.
<http://webpages.shepherd.edu/amihailo>

Moore, Gregory A. 2001. *Integrating Maple Into the Math Curriculum, A Sensible Guide for Educator*. Waterloo, Waterloo Maple.Inc.

Waterloo Maple Inc. 1999. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Winter 1998-99*

Waterloo Maple Inc. 2001a. *Maple 7 Online Help*. <http://www.maplesoft.com/>

Waterloo Maple Inc. 2001b. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Spring 2001; Fall 2001;*

Waterloo Maple Inc. 2002. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Winter 2002*